

Sample SPICE deck using BIRD 122 Generic and Broadband models

Prepared by Walter Katz, Signal Integrity Software, Inc.

December 14, 2010

SPICE deck that represents one way of generating a step response simulation when the user specifies either a Generic Analog Model or Broadband Analog Model as specified in BIRD 122.

The following spice deck is totally generic; all elements are IBIS ISS elements except for the elements V_H_stim and V_L_stim which simply generate step responses between Vol and Voh.

Red are generic model elements
Blue are Tstonefile model elements

Tx and Rx Touchstone files are not provided. EDA Vendors can get exemplar Touchstone files directly from IC Vendors under NDA where appropriate.

```
* Start of Step Response SPICE deck
*
*
.param Voh=1.
.param Vol=0.
.param Tx_Rs=50.
.param Tx_Cc=.950f
.param Tx_Rt=1Meg
.param Tx_Vt=0.
.param Tx_Rd=1Meg
.param Tx_Cd=1f
.param Tx_Trf=10p
.param Tx_Trf=1f
* Stimulus
* step starts at 1ns
V_H_stim H_stim 0 DC='Vol'
+      pwl( 0 'Vol' 1n 'Vol' '1n+Tx_Trf/.6' 'Voh' 100n 'Voh')
V_L_stim L_stim 0 DC='Voh'
+      pwl( 0 'Voh' 1n 'Voh' '1n+Tx_Trf/.6' 'Vol' 100n 'Vol')
*
*
* Tx Generic Model
*
* Rs
RTx_Rs_H H_stim Tx_H_pad 'Tx_Rs'
RTx_Rs_L L_stim Tx_L_pad 'Tx_Rs'
*CC
CTx_Cc_H Tx_H_pad 0 'Tx_Cc'
CTx_Cc_L Tx_L_pad 0 'Tx_Cc'
* Rt/Vt
```

```

RTx_Rt_H Tx_H_pad Tx_Vt 'Tx_Rt'
RTx_Rt_L Tx_L_pad Tx_Vt 'Tx_Rt'
VTx_Rt Tx_Vt 0. DC='Tx_Vt'
* Rd
RTx_Rd Tx_H_pad Tx_L_pad 'Tx_Rd'
* Cd
CTx_Cd Tx_H_pad Tx_L_pad 'Tx_Cd'
*
* Tx Tstonefile Model
*
STx H_stim Tx_H_pad L_stim Tx_L_pad 0 mname=Tx_typ
.model Tx_typ S N=4 TSTONEFILE='Tx.s4p'
*
* the channel
*
W1 Tx_H_PAD Tx_L_PAD GND Rx_H_PAD Rx_L_PAD GND N=2
+ RLGCmodel=Channel_Wline L=0.5080
.model Channel Wline W MODELTYP=RLGC N=2
+ Ro = +1.02785E+01 +0.00000E+00 +1.02785E+01
+ Lo = +4.37200E-07 +9.24700E-08 +4.37200E-07
+ Co = +1.13200E-10 -2.39500E-11 +1.13200E-10
+ Rs = +1.64438E-03 +7.28589E-05 +1.64438E-03
+ Gd = +1.42300E-11 -3.01000E-12 +1.42300E-11
*
* Rx Generic Model
*
* Cc
.param Rx_Cc=.950f
.param Rx_Rt=50.
.param Rx_Vt=0.
.param Rx_Rd=1Meg
.param Rx_Cd=1f
CRx_Cc_H Rx_H_pad 0 'Rx_Cc'
CRx_Cc_L Rx_L_pad 0 'Rx_Cc'
* Rt/Vt
RRx_Vt_H Rx_H_pad Rx_Vt 'Rx_Rt'
RRx_Vt_L Rx_L_pad Rx_Vt 'Rx_Rt'
VRx_Vt Rx_Vt 0 DC='Rx_Vt'
* Rd
RRx_Rd Rx_H_pad Rx_L_pad 'Rx_Rd'
* Cd
CRx_Cd Rx_H_pad Rx_L_pad 'Rx_Cd'
*
* The impulse response is the first derivative of v(Rx_H_pad-Rx_L_pad)
*
* Rx Tsdtonefile model
*
SRx Rx_H_pad Rx_H_z Rx_L_pad Rx_L_z 0 mname=Rx_typ
.model Rx_typ S N=4 TSTONEFILE='Rx.s4p'
*
RRx_H_z Rx_H_z 0 50
RRx_L_z Rx_L_z 0 50

```

```
*  
* The impulse response is the first derivative of v(Rx_H_z - Rx_L_z)  
*  
.END
```